

Carlos Roberto Dos Santos Junior, N<sup>o</sup>USP 9435102  
William Luis Alves Ferreira, N<sup>o</sup>USP: 9847599

## **Experiência 4**

Universidade de São Paulo – USP  
Escola de Engenharia de São Carlos – EESC  
Instituto de Ciências Matemáticas e de Computação – ICMC  
Programa de Graduação

Brasil

2021

# Sumário

<b>1</b>	<b>INTRODUÇÃO</b>	<b>3</b>
<b>2</b>	<b>QUESTÕES</b>	<b>4</b>
Questão 1		4
Questão 2		4
Questão 4		5
Questão 5		11
Questão 6		11
Questão 7		14
Questão 8		18
Questão 9		19
Questão 10		20
Questão 11		21
Questão 12		21
Questão 13		25
Questão 14		27
Questão 15		31
Questão 16		33
Questão 17		33

# 1 Introdução

Assim como no experimento 3, neste documento aprofundaremos os conhecimentos no desenvolvimento de layouts, produzindo circuitos como um oscilador em anel, também aprenderemos um novo conceito que é a capacidade de extrair o layout considerando os elementos como capacitâncias, resistências e elementos parasitas, essa extração também facilita o processo de simulação desse circuito.

Vamos aprofundar também conceitos de atraso de propagação na saída de inversores, o comportamento dos inversores, consumos de potência.

## 2 Questões

**Questão 1:** Desenhe um inversor com  $W_n = \underline{\hspace{2cm}}$  e verifique com o DRC.


Partindo de  $W_n = 2,03\mu m \approx 2,05\mu m$ , temos que calcular nosso  $W_p$  para podermos desenhar o nosso inversor. Utilizando a relação encontrada no relatório 2, temos:

$$W_p = 2,9365 \times W_n$$

$$W_p = 6,019825 \approx 6,00\mu m$$

Com esses valores, conseguimos então desenhar o inversor que segue na figura 1.

Figura 1 – Inversor com  $W_n = 2,05\mu m$  e output do DRC.



Fonte: Pelos próprios autores

**Questão 2:** Na opção Outputs - Extraction type, há varias opções para extração (C+CC, R, R+C, R+C+CC). O que significa cada um destes tipos de extração?

Esses tipos de extração referem-se aos tipos de componentes parasitas que o software deve levar em consideração.

- C+CC - Considerado dois tipos de capacitâncias, as capacitâncias de acoplamento e as capacitâncias que aparecem entre o substrato e os materiais que compõem as trilhas.
- R - Considerado as resistências presentes no circuito.
- R+C - Considerado as resistências e as capacitâncias que aparecem entre o substrato e os materiais que compõem as trilhas.
- R+C+CC - Considerado as resistências, capacitâncias de acoplamento e as capacitâncias que aparecem entre o substrato e os materiais que compõem as trilhas.

**Questão 4:** \*Extraia com a opção C+CC. Prepare o arquivo para simulação e simule o circuito com parâmetros típicos. Determine a máxima frequência de operação, os atraso na propagação (para subida e descida) e consumo de potência do circuito. Considere:

- Capacitância de saída de 50fF;
- VDD= 3,0 V;
- o sinal de entrada com (tempo de subida) = (tempo de descida) = 1% do período. Certifique-se de que a frequência escolhida é conveniente para o que deseja

Primeiro, para encontrar a máxima frequência de operação realizaremos um excursionamento de frequência com o atributo **SWEEP** do comando **TRAN**, assim realizando uma análise transitória com excursionamento de 1 a 3,5GHz. Segue trechos do arquivo netlist extraído com PEX C+CC acrescido dos comandos **MEAS**:

```
*** Parâmetros de Projeto.
```

```
.Param F=1G P='1/F'
```

```
*** 1% de tempo de propagacao de subida e decida
```

```
Vin in 0 PULSE(0 3.0 0 '0.01*P' '0.01*P' '0.49*P' P)
```

```
*** Medidas Por Escursionamento
```

```
*** A - Operacao para encontrar maxima frecuencia de operacao (Do enunciado)
```

```
.meas tran MinZero find v(out) when v(in)=2.9 fall=5
```

```
.meas tran MaxUm find v(out) when v(in)=0.1 rise=5
```


```

** Comando tran analise transitoria
*.tran <TPrint> <TSTOP> <TSTART> <SWEEP> <UIC> <MONTE>
*** Escursionando frequencia em busca da maxima
.tran P '6*P' 0 'P/1000' sweep F INCR 0.1G 1G 3.5G
.probe tran ALL

```

A seguir na figura 2, temos as frequências relevantes considerando aproximadamente 3% da tensão máxima (3V) e mínima (0V), respectivamente, 2,9V e 0,1V.

Figura 2 – Curvas das funções *MEANS* com as tensões variantes da analise transistórica com excursionamento de frequência de operação de 1G a 3,5GHz



Fonte: Pelos próprios autores

Na figura 2 temos a marcação das frequências 1,55 e 1,69GHz correspondente as delimitações de 2,9V como tensão máxima e 0,1V mínimo. Por fim, temos a frequência máxima de operação 1,55 GHz, considerando um desvio tolerável de 0,1V.

Após encontrar frequência máxima através das medidas extraídas **MinZero** e **MaxUm**, temos o cálculo da potência e propagação de atraso de descida e subida, através do arquivo netlist completo a seguir:

```

* File: inversor.pex.netlist
* Created: Thu Aug 5 22:23:24 2021
* Program "Calibre xRC"
* Version "v2006.2_16.16"
*
*** Configuracao

```

```
.option measDGT=8
.options ingold=1
.option lis
.option hmax=0.01n

*** Parametros de Projeto.
*.Param F=1G P='1/F'
*** Frequencia Maxima de operacao
.Param F=1.5446G P='1/F'

*** Circuito extraido
.subckt INVERSOR VSS VIN VOUT VDD
*
mM0 VOUT VIN VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM1 VOUT VIN VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
c_4 VSS 0 0.229412f
c_8 VDD 0 19.5184f
c_12 VIN 0 0.549408f
c_16 VOUT 0 0.329072f

*
.include "inversor.pex.netlist.INVERSOR.pxi"
*
.ends INVERSOR
.include Model35_Eldo

*** Tensoes estabelecidas como base
Vdd vd 0 DC 3V
*** Conectando subcircuito
Cout out 0 50fF
Xinv 0 in out vd INVERSOR

*** 1% de tempo de propagacao de subida e decida
Vin in 0 PULSE(0 3.0 0 '0.01*P' '0.01*P' '0.49*P' P)

*** Medidas Por Escursionamento
```

```
*** A - Operacao para encontrar máxima frequência de operação (Do enunciado)
.meas tran MinZero find v(out) when v(in)=2.9 fall=5
.meas tran MaxUm find v(out) when v(in)=0.1 rise=5

*** B - Tempo de propagacao de descida e subida
.meas tran tempodecida trig v(in) val=1.5 rise=5 targ v(out) val=1.5 fall=5
.meas tran temposubida trig v(in) val=1.5 fall=5 targ v(out) val=1.5 rise=5

*** Calculo da potencia
*** testando função potencia
.meas tran pot AVG P(Vdd) from='P' to='6*P'


*** Calculado
.meas tran correnteavg AVG I(Vdd) from='P' to='6*P'
**Com a media da corrente podemos utilizar VDD=3 dado no enunciado
.meas tran potcalc param='-correnteavg*3'

** Comando tran analise transitoria
*.tran <TPrint> <TSTOP> <TSTART> <SWEEP> <UIC> <MONTE>
*** Escursionando frequencia em busca da maxima
*.tran P '6*P' 0 'P/1000' sweep F INCR 0.1G 1G 3.5G
*** transitória para a frequência de operação esperada
.tran '5*P' '10*P' '5*P' '10*P'
.probe tran ALL
.end
```

Para encontrar os tempos de propagação de atraso de subida e descida utilizamos os comandos **.MEAS** com o *trig* na mediana da curva para tensão de 3V de pico para entrada **PULSE**, assim extraíndo a curva do tempo de propagação de atraso, como visto na figura 3. Com isso, podemos verificar o tempo de propagação do atraveso de subida e descida, respectivamente é 94,82 e 111,90 ps para a frequência máxima de operação.



Figura 3 – Curvas de tempo de propagação de descida e subida marcadas para as frequências encontradas na figura 2




Fonte: Pelos próprios autores

Com o comando a seguir temos a curva de consumo de potência pelo circuito (presente na figura 4), marcado para a frequência máxima de operação de 1,55GHz, correspondendo ao consumo de 0,874 mW. Visto que temos a função potência média para o nó **Vdd** para extrair a curva de consumo, utilizamos o cálculo da potência com a orientação de corrente adequada para representar o consumo de potência, pelo cálculo “-correnteavg\*3”.

```
*** Calculo da potencia
*** testando função potencia
.meas tran pot      AVG P(Vdd) from='P' to='6*P'
*** Calculado
.meas tran correnteavg AVG I(Vdd) from='P' to='6*P'
**Com a media da corrente pordemos utilizar VDD=3 dado no enunciado
.meas tran potcalc param='-correnteavg*3'
```


Figura 4 – Curva de consumo de potência no excursionamento de frequência com marcação na frequência máxima de 1,55GHz



Fonte: Pelos próprios autores

Para concluir a análise do circuito apresentamos a tensão de entrada e saída na figura 5.

Figura 5 – Curvas das tensões de entrada e saída para a frequência máxima de 1,55GHz




Fonte: Pelos próprios autores

**Questão 5:** Observe no ezwave as correntes que passam pelos transistores. Coloque no relatório o gráfico das correntes junto com a as tensões e justifique o comportamento das correntes.

Na figura 6 temos o gráfico pedido, neste gráfico o inversor com o nome MM0 é referente ao NMOS, enquanto MM1 ao PMOS. Podemos ver que quando temos uma tensão de entrada em seu máximo, o transistor NMOS gera um pico na corrente de dreno, essa corrente diminui a medida que as tensões entre o nó de saída e o terra se aproximam, quando essas tensões se igualam a corrente no dreno do NMOS é cortada. Quando temos a tensão de entrada em seu mínimo o transistor PMOS começa a conduzir corrente em seu dreno.

Figura 6 – Gráficos contendo as correntes que passam pelos transistores PMOS E NMOS em paralelo com as tensões de entrada e saída.




Fonte: Pelos próprios autores

**Questão 6:** Extraia agora com a opção R+C+CC. Prepare o arquivo para simulação e simule o circuito com parâmetros típicos. Determine a máxima frequência de operação, os atraso na propagação (para subida e descida) e consumo de potência do circuito. Compare com o resultado anterior (o quanto piorou e se houve aumento expressivo no tempo de simulação).

Replicaremos aqui as análises realizadas na questão 4 para extração R+C+CC, portanto, resumidamente, temos que a máxima frequência de operação é 1,50GHz (presente na figura 7), com potência consumida de 0,851 mW (presente na figura 9) e tempos de


atraso de propagação de subida e descida, respectivamente, 96,94 e 113,22 ps (presente figura 8), por fim, curva de entrada e saída de tensão para a frequência máxima de operação, exposto na figura 10.

Figura 7 – Curvas das funções *MEANS* com as tensões variantes da análise transistórica com excursionamento de frequência de operação de 1G a 3,5GHz




Fonte: Pelos próprios autores

Figura 8 – Curvas de tempo de propagação de descida e subida marcadas para as frequências encontradas na figura 7




Fonte: Pelos próprios autores

Figura 9 – Curva de consumo de potência no excursionamento de frequência com marcação na frequência máxima de 1,50GHz



Fonte: Pelos próprios autores

Figura 10 – Curvas das tensões de entrada e saída para a frequência máxima de 1,50GHz



Fonte: Pelos próprios autores

Como resultado temos a tabela 1, com o sumário das análises realizadas com extração C+CC e R+C+CC. Verificamos que a frequência máxima de operação e potência consumida teve queda de aproximadamente 3% e aumento de tempo de atraso de propagação de subida e descida de aproximadamente 2%, já para o tempo de simulação como o acréscimo de componente é ínfimo frente aos demais cálculos envolvidos na simulação.

Tabela 1 – Tabela comparando as características do circuito com extração C+CC e R+C+CC

Característica	Extração C+CC	Extração R+C+CC	Unidade
Frequência Máxima	1,55	1,50	GHz
Potência Consumida	0,874	0,851	mW
Atraso Prop. Subida	94,82	96,94	ps
Atraso Prop. descida	111,90	113,22	ps
Tempo de Simulação	1	1	s

Fonte: Pelos próprios autores

**Questão 7:** Para o circuito extraído com a opção C+CC determine os atrasos na propagação (para subida e descida) e consumo de potência do circuito para capacitâncias de carga de 50 fF, 100fF, 150 fF, 200 fF, 250 fF e 300 fF (certifique-se que o período do sinal de entrada é conveniente, que o tempo de simulação é suficiente e que o HMAX dê a precisão desejada). Mantenha no sinal de entrada tempo de subida/descida de 1% do período. (inclua no relatório linhas de comando e de sinais de entrada do arquivo de simulação e figura com os sinais de entrada/saída para carga de 300fF)

Para este item temos o excursionamento da capacitância acoplada na saída de 50 a 300fF com incremento de 50fF. Abaixo segue as linhas de comando do arquivo de simulação:

```
*** Configuracao
.option measDGT=8
.options ingold=1
.option lis
*.option hmax=0.01n

*** Parâmetros de Projeto.
.Param F=0.1G P='1/F'

*** Escursionamento de capacitancia
.Param Ccout=50fF

*** Circuito extraido
.subckt INVERSOR VSS VIN VOUT VDD
*
mMO VOUT VIN VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
```

```

mM1 VOUT VIN VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
c_4 VSS 0 0.229412f
c_8 VDD 0 19.5184f
c_12 VIN 0 0.549408f
c_16 VOUT 0 0.329072f

*
.include "inversor.pex.netlist.INVERSOR.pxi"
*
.ends INVERSOR
.include Model35_Eldo

*** Tensoes estabelecidas como base
Vdd vd 0 DC 3V
*** Conectando subcircuito
Cout out 0 Ccout
Xinv 0 in out vd INVERSOR

*** Sinal de entrada
*** 1% de tempo de propagacao de subida e decida
Vin in 0 PULSE(0 3.0 0 '0.01*P' '0.01*P' '0.49*P' P)

*** Medidas Por Escursionamento

*** A - Operacao para encontrar maxima frequencia de operacao (Do enunciado)
.meas tran MinZero find v(out) when v(in)=2.9 fall=5
.meas tran MaxUm find v(out) when v(in)=0.1 rise=5

*** B - Tempo de propagacao de descida e subida
.meas tran tempodecida trig v(in) val=1.5 rise=5 targ v(out) val=1.5 fall=5
.meas tran temposubida trig v(in) val=1.5 fall=5 targ v(out) val=1.5 rise=5

*** Calculo da potencia
.meas tran pot AVG P(Vdd) from='P' to='6*P'
.meas tran correnteavg AVG I(Vdd) from='P' to='6*P'

```

```

**Com a media da corrente pordemos utilizar VDD=3 dado no enunciado
.meas tran potcalc param='-correnteavg*3'

** Comando tran analise transitoria
*.tran <TPrint> <TSTOP> <TSTART> <SWEEP> <UIC> <MONTE>
*** Escursionando frecuencia em busca da maxima
*.tran P '6*P' 0 'P/1000' sweep F INCR 0.1G 1G 3.5G
*** Escursionando capacitancia de saida
.tran P '6*P' 0 'P/100' sweep Cout INCR 50fF 50fF 300fF
*** transitoria para a frecuencia de operacao esperada
*.tran '5*P' '10*P' '5*P' '10*P'
.probe tran ALL

.end

```

Utilizando a simulação acima conseguimos obter os atrasos de propagação e consumo de potência para as capacitâncias de carga pedidas, esses valores podemos ver na tabela 2.

Tabela 2 – Tabela com o consumo de potências e atrasos de descida/subida para cada capacitância de carga


Capacitância	Cons. potência [mW]	Atr. descida [ps]	Atr. subida [ps]
50fF	0,060	134,12	116,02
100fF	0,105	219,15	190,83
150fF	0,150	302,13	264,75
200fF	0,195	385,18	339,37
250fF	0,240	468,97	413,13
300fF	0,285	553,04	486,21

Fonte: Pelos próprios autores

Podemos ver na figura 11 os sinais de entrada/saída para a carga de 300fF.




Figura 11 – Gráfico com sinais de entrada/saída para a carga de 300fF.



Fonte: Pelos próprios autores

e , conferimos a discrepância na figura 12 os sinais de entrada/saída para a carga de 50fF.

Figura 12 – Gráfico com sinais de entrada/saída para a carga de 50fF.



Fonte: Pelos próprios autores


**Questão 8:** Trace, com os resultados anteriores, o gráfico atraso (subida e descida) X carga de saída. Determine a inclinação dessa curva (não esqueça de fornecer as dimensões).

Partindo dos resultados anteriores, conseguimos através da ferramenta EZWaze gerar o gráfico 13, através da funcionalidade *slope* podemos determinar a inclinação dessas curvas, com isso temos:

$$Inclinacao_{subida} = 1,49ps/fF$$

$$Inclinacao_{descida} = 1,66ps/fF$$

Figura 13 – Gráfico de atraso (subida e descida) X carga de saída




Fonte: Pelos próprios autores

**Questão 9:** Trace o gráfico potência consumida X carga de saída. Determine a inclinação dessa curva (não esqueça de fornecer as dimensões).

Conseguimos através da ferramenta EZWaze gerar o gráfico 14, através da funcionalidade *slope* podemos determinar a inclinação da curva no gráfico de potência X carga de saída, com isso temos:

$$Inclinacao_{potencia} = 0,901mW/fF$$

Figura 14 – Gráfico de potência consumida X carga de saída




Fonte: Pelos próprios autores

**Questão 10:** Obtenha com os colegas da classe as inclinações das curvas encontradas nos itens 8 e 9 e trace os gráficos inclinação X  $W_n$ .


Os dados foram obtidos, e foi possível gerar os gráficos que estão presentes nas figuras 15 e 16.

Figura 15 – Gráfico de inclinação do tempo de subida e descida X  $W_n$  para 3V.



Fonte: Pelos próprios autores

Figura 16 – Gráfico de inclinação de potência consumida X  $W_n$  para 3V.



Fonte: Pelos próprios autores

**Questão 11:** Justifique o formato das curvas dos itens 8, 9 e 10 (relações úteis estão no livro do Sedra, capítulo 5).

Através da referência dada no enunciado, conseguimos as 3 relações abaixo:

$$t_{PLH} = \frac{1,6C_L}{C_{ox}\mu_p \frac{W_p}{L} V_{DD}}$$

$$t_{PHL} = \frac{1,6C_L}{C_{ox}\mu_n \frac{W_n}{L} V_{DD}}$$

$$Pot = \frac{1}{T} \int_T iV_{DD} dt = \frac{1}{T} C_L V_{DD}^2$$

Como podemos ver nas duas primeiras relações, temos que os atrasos de propagação de descida/subida são diretamente proporcionais a  $C_L$  (capacitância de carga), com isso podemos explicar o comportamento do gráfico da figura 13 da questão 8, vemos também nessas relações que os atrasos de propagação são inversamente proporcionais a  $W_n$  e  $W_p$ .

Analisando a última relação, a relação de potência, podemos ver que o consumo de potência é diretamente proporcional ao valor da capacitância de carga, o que nos ajuda a ver o comportamento do gráfico da figura 14 da questão 9. Através da mesma relação podemos ver que o consumo de potência não depende de  $W_n$ , então não há variação com a mudança desse parâmetro, explicando assim a figura 16 da questão 10.

**Questão 12:** Repita os itens 7, 8 e 9 utilizando agora tensão de alimentação de 2,0 V. Compare os resultados com os anteriores e justifique as diferenças.

Fazendo os mesmos passos anteriores porém modificando a tensão de alimentação para 2V teremos:

\*\*\* Sinal de entrada

\*\*\* 1% de tempo de propagacao de subida e decida

Vin in 0 PULSE(0 2.0 0 '0.01\*P' '0.01\*P' '0.49\*P' P)

\*\*\* A - Operacao para encontrar maxima frequencia de operacao (Do enunciado)

.meas tran MinZero find v(out) when v(in)=1.9 fall=5

.meas tran MaxUm find v(out) when v(in)=0.1 rise=5

\*\*\* B - Tempo de propagacao de descida e subida

.meas tran tempodecida trig v(in) val=1 rise=5 targ v(out) val=1 fall=5

.meas tran temposubida trig v(in) val=1 fall=5 targ v(out) val=1 rise=5

\*\*\* Calculo da potencia

```
.meas tran pot      AVG P(Vdd) from='P' to='6*P'
.meas tran correnteavg AVG I(Vdd) from='P' to='6*P'
**Com a media da corrente pordemos utilizar VDD=2 dado no enunciado
.meas tran potcalc param='-correnteavg*2'
*** Escursionando capacitancia de saida
.tran P '6*P' 0 'P/100' sweep Cout INCR 50fF 50fF 300fF
```


Com a mudança de 3V para 2V na tensão de alimentação, conseguimos ver diversas diferenças nos resultados. Comparando os resultados da tabela 2 e da figura 13 com a tabela 3 e a figura 17 podemos ver que os tempos de propagação, tanto de subida quanto descida estão maiores, isso ocorre porque esses tempos são inversamente proporcionais à tensão de alimentação.

Tabela 3 – Tabela com o consumo de potências e atrasos de descida/subida para cada capacitância de carga para alimentação de tensão de 2V

Capacitância	Cons. potência [mW]	Atr. descida [ps]	Atr. subida [ps]
50fF	0,04	186,27	181,90
100fF	0,07	304,68	302,40
150fF	0,10	424,11	422,52
200fF	0,13	542,66	543,31
250fF	0,16	663,78	661,16
300fF	0,19	784,12	779,54

Fonte: Pelos próprios autores

Figura 17 – Gráfico de atraso (subida e descida) X carga de saída




Fonte: Pelos próprios autores

Sendo inclinação para o tempo de atraso de propagação de subida de 2,37 ps/fF e descida de 2,42 ps/fF.

É possível observar também, comparando as figuras 14 e 18 que houve um queda na potência consumida para uma inclinação na relação potência e capacitância de 0,400 mW/fF, isso ocorre porque a potência consumida é diretamente proporcional à tensão de alimentação.


Figura 18 – Gráfico de potência consumida X carga de saída



Fonte: Pelos próprios autores

Verificamos também que a tensão de saída em resposta a entrada possui curva semelhante a alimentação de 3V, com visto na figura 19


Figura 19 – Gráfico com sinais de entrada/saída para a carga de 300fF para alimentação 2V.



Fonte: Pelos próprios autores


Por fim, temos os dados de inclinação do consumo de potencia e tempo de atraso de propagação de descida e subida, idem a questão 10, como visto nas figuras 20 e 21.

Figura 20 – Gráfico de inclinação do tempo de subida e descida por capacitância X  $W_n$  para 2V.



Fonte: Pelos próprios autores

Figura 21 – Gráfico de inclinação de potência consumida por capacitância X  $W_n$  para 2V.



Fonte: Pelos próprios autores

Nota-se que as inclinações sobre capacitância para consumo de potência pela




dimensão de  $W_n$  continua constante nesta tensão de alimentação com relação de 4, e , assim como relação exponencial da inclinação sobre capacitância dos tempos de atraso de propagação de descida e subida mantém uma relação exponencial com o valor de  $W_n$  tendo valor inicial aproximadamente 0,75 unidades a mais devido a relação vista na questão 11, assim como a diminuição a relação do consumo de potência e capacitância acoplada na saída, visto a alimentação de 3V para 2V.

**Questão 13:** Desenhe no ICSTATION 3 inversores, com as dimensões utilizadas no item 1, um conectado ao outro. Passe o DRC e faça a extração C+CC. Aplique uma onda quadrada à entrada do circuito extraído e meça os atrasos na propagação (subida e descida) do inversor do meio.

Utilizando a ferramenta *ICSTATION* e com as dimensões de  $W_n = 2,05\mu m$  e  $W_p = 6\mu m$  conseguir desenhar o inversor da figura 22.

Figura 22 – Gráfico de potência consumida X carga de saída



Fonte: Pelos próprios autores

Após feita a extração C+CC, fizemos todas as modificações necessárias no *netlist* para podermos aplicar a onda quadrada para obter os atrasos de propagação, abaixo temos o *netlist* utilizado e a tabela 4 possui os valores de atraso de propagação observados.

```
*** Configuracao
.option measDGT=8
.options ingold=1
.option hmax=0.01n
```

```

*** Parametros de Projeto.
.Param F=1G P='1/F' Rval=1
*
.subckt TRESINVERSORES  VSS VIN VDD VIN2 VOUT2
*
mM0 VIN2 VIN VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM1 VOUT2 VIN2 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM2 VOUT VOUT2 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM3 VIN2 VIN VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM4 VOUT2 VIN2 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM5 VOUT VOUT2 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
c_4 VIN 0 0.549408f
c_9 VIN2 0 1.12989f
c_15 VSS 0 2.52472f
c_21 VDD 0 42.0457f
c_26 VOUT2 0 1.12989f
c_30 VOUT 0 0.329072f
*
.include "tresinversores.pex.netlist.TRESINVERSORES.pxi"
*
.ends TRESINVERSORES
.include Model35_Eldo

*** Conectando circuito
Xinv vs in vd VIN2 VOUT2 TRESINVERSORES
*
*
*** Tensoes estabelecidas como base
Vin in 0 PULSE(0 3.0 0 '0.01*P' '0.01*P' '0.49*P' P)
Vdd vd 0 DC 3V
Vss vs 0 DC 0
Rt vs 0 Rval *** NAO copiar essa linha

```

```

**** Inversor 2
.meas tran tempodecida_INV2 trig v(VIN2) val=1.5 rise=5
    targ v(VOUT2) val=1.5 fall=5
.meas tran temposubida_INV2 trig v(VIN2) val=1.5 fall=5
    targ v(VOUT2) val=1.5 rise=5
*
*
*** Simulacao
.tran '0*P' '10*P' '0*P' 'P/1000'

*** Comando IC - estabelecimento valor inicial para entrada VIN 3V
.probe tran all

.end

```

Tabela 4 – Tabela com atrasos de descida/subida do circuito contendo 3 inversores.

Atraso na descida	Atraso na subida
78,28ps	65,08ps

Fonte: Pelos próprios autores


Na figura 23 podemos observar os sinais de entrada/saída do segundo inversor presente no circuito mostrado acima.

**Questão 14:** Desenhe no ICSTATION 5 inversores iguais e conecte a saída de um a entrada de outro; a saída do último inversor deve ser conectada a entrada do primeiro. Este circuito é chamado de oscilador em anel com 5 inversores. Passe o DRC e faça a extração C+CC. Simule o circuito e determine o período de oscilação (obs. talvez tenha de utilizar o .ic para impor condições iniciais de simulação. Qual é a função desse comando).

Primeiramente desenhamos o circuito pedido na questão e passamos o DRC para verificar os erros, na figura 24 temos o circuito juntamente com a saída do DRC.


Após isso, conseguimos extrair o circuito utilizando a extração C+CC e fizemos a simulação, obtemos os períodos de 740,11ps para 3V de entrada e 1204,7ps para 2V de entrada, conforme vemos no gráfico da figura 25.

Figura 23 – Gráfico com os sinais de entrada e saída do segundo inversor.



Fonte: Pelos próprios autores

Figura 24 – Layout do circuito oscilador em anel com 5 inversores ao lado do DRC.




Fonte: Pelos próprios autores

Podemos ver que nosso valor ficou abaixo do esperado, para melhorarmos isso, iremos utilizar o METAL 2 na formação de nosso oscilador em anel, colocando-o para unir a saída do último inversor com a entrada do primeiro inversor. Após essa modificação, obtemos o circuito da figura 26.

Realizamos novamente a extração e através das linhas de comando abaixo, foi possível obter a figura 27.


```
*** Configuracao
.option measDGT=8
```

Figura 25 – Gráfico com o período de oscilação gerado pelo circuito para uma tensão de alimentação de 3V e de 2V.



Fonte: Pelos próprios autores

Figura 26 – Layout do circuito oscilador em anel com 5 inversores com MET2 ao lado do DRC.



Fonte: Pelos próprios autores

```
.options ingold=1
.option hmax=0.01n
```

```
*** Parametros de Projeto.
.Param F=1G P='1/F'
```

```
*** Circuito extraido
.subckt OSCILADOR VSS VIN VDD
*
```

```
mM0 2 VIN VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
```

```

mM1 3 2 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM2 4 3 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM3 7 4 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM4 VIN 7 VSS VSS MODN L=3.5e-07 W=2.05e-06 AD=1.7425e-12 AS=1.9475e-12
+ PD=3.75e-06 PS=3.95e-06 NRD=0.207317 NRS=0.207317
mM5 2 VIN VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM6 3 2 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM7 4 3 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM8 7 4 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
mM9 VIN 7 VDD VDD MODP L=3.5e-07 W=6e-06 AD=5.1e-12 AS=5.7e-12 PD=7.7e-06
+ PS=7.9e-06 NRD=0.0708333 NRS=0.0708333
c_7 VIN 0 1.63397f
c_12 2 0 1.11989f
c_18 3 0 1.1281f
c_24 4 0 1.1281f
c_31 VSS 0 4.32547f
c_38 VDD 0 64.2998f
c_43 7 0 1.1281f
*
.include "oscilador.pex.netlist.OSCILADOR.pxi"
*
.ends OSCILADOR
.include Model35_Eldo

*** Conectando circuito
Xinv vs in vd OSCILADOR
*
*
*** Tensoes estabelecidas como base
Vdd vd 0 DC 3V
Vss vs 0 DC 0

```

```


*** Simulacao
.tran '0*P' '10*P' '0*P' 'P/1000' UIC

*** Comando IC - estabelecimento valor inicial para entrada VIN 3V
.ic v(in)=3V
.probe tran v(in)

.end

```

Figura 27 – Gráfico com o período de oscilação gerado pelo circuito com MET2 para uma tensão de alimentação de 3V.



Fonte: Pelos próprios autores

Vemos no gráfico da figura 27 que o valor do período de oscilação é de 733,13ps.

**Questão 15:** Qual é a relação entre o período de oscilação de um oscilador em anel e os atrasos na propagação dos inversores? Verifique para seu circuito se a relação é obedecida.

O período de oscilação de um oscilador em anel é determinado por:

$$T = \sum_{i=1}^n (t_{PHL} + t_{PLH})$$

onde  $n$  é o número de inversores.

Para verificarmos se nosso circuito obedece essa relação, utilizamos os comandos abaixo para podermos calcular os tempos de propagação de cada inversor.

```

*** Tempo de propagacao de descida e subida
**** Inversor 1
.meas tran tempodecida_INV1 trig v(in) val=1.5 rise=5 targ v(VOUT1)
    val=1.5 fall=5
.meas tran temposubida_INV1 trig v(in) val=1.5 fall=5 targ v(VOUT1)
    val=1.5 rise=5
**** Inversor 2
.meas tran tempodecida_INV2 trig v(VOUT1) val=1.5 rise=5 targ v(VOUT2)
    val=1.5 fall=5
.meas tran temposubida_INV2 trig v(VOUT1) val=1.5 fall=5 targ v(VOUT2)
    val=1.5 rise=5
**** Inversor 3
.meas tran tempodecida_INV3 trig v(VOUT2) val=1.5 rise=5 targ v(VOUT3)
    val=1.5 fall=5
.meas tran temposubida_INV3 trig v(VOUT2) val=1.5 fall=5 targ v(VOUT3)
    val=1.5 rise=5
**** Inversor 4
.meas tran tempodecida_INV4 trig v(VOUT3) val=1.5 rise=5 targ v(VOUT4)
    val=1.5 fall=5
.meas tran temposubida_INV4 trig v(VOUT3) val=1.5 fall=5 targ v(VOUT4)
    val=1.5 rise=5
**** Inversor 5
.meas tran tempodecida_INV5 trig v(VOUT4) val=1.5 rise=5 targ v(in)
    val=1.5 fall=5
.meas tran temposubida_INV5 trig v(VOUT4) val=1.5 fall=5 targ v(in)
    val=1.5 rise=5

```

Obtemos assim a tabela 5 e aplicando a relação obtemos  $T = 715,68ps$ , o que é próximo ao período calculado na questão anterior, que é de  $733,13ps$ , então podemos afirmar que nosso circuito de fato obedece essa relação.

Tabela 5 – Tabela com atrasos de descida/subida do circuito em anel contendo 5 inversores.

Inversor	Atraso na descida	Atraso na subida
1	79,25ps	69,38ps
2	76,69ps	66,81ps
3	76,78ps	66,60ps
4	76,68ps	66,43ps
5	72,00ps	65,06ps


Fonte: Pelos próprios autores



**Questão 16:** Obtenha com os colegas da classe os períodos que eles obtiveram e trace o gráfico Período X  $W_n$ . Justifique o formato da curva obtida.

Os dados foram obtidos com os colegas da classe, e foi possível gerar o gráfico da figura 28.

Figura 28 – Gráfico com o período(ns) X  $W_n(\mu m)$  para 3V.



Fonte: Pelos próprios autores

Na figura 28 vemos que a medida que o valor de  $W_n$  aumenta, o valor do período diminui aproximadamente de maneira exponencial, tendendo a uma estabilização (plato) em torno de  $0,67ns$ .


**Questão 17:** Determine agora o período de oscilação utilizando tensão de alimentação de 2 V. Compare o resultado com o anterior e justifique as diferenças.

Iremos primeiramente modificar o arquivo de simulação para que possamos obter através da ferramenta *EZWAVE* o que precisamos, então modificamos as seguintes linhas do arquivo:

```
Vdd vd 0 DC 2V
.ic v(in)=2V
```

Assim obtemos o gráfico da figura 29, onde podemos ver que o período de oscilação é de  $1194,5\mu s$ , vemos então uma grande mudança nesse valor, isso ocorreu porque os tempos de atraso são inversamente proporcionais à tensão de alimentação, causando assim um aumento no período de oscilação.

Figura 29 – Gráfico com a oscilação gerada pelo circuito para uma tensão de 2V.



Fonte: Pelos próprios autores